



Ing. Jiri Otyka, CSc

Calculating the Focal Point of an Offset Dish Antenna

It is assumed that readers know how to calculate the focal point of a symmetrical parabolic antenna.

Offset antennae are a different matter. The popular belief is that the focal point cannot be established by simple means. However, this is not correct.

To establish the focal point of an offset antenna, we need to measure three parameters instead of two: the large and small elliptical axes and the maximum antenna depth (T in the basic program, h on the diagrams).

An offset antenna is formed by a three dimensional space shown in Fig 1. The "similarity" of offset antennae is determined by two parameters shown in Fig 2, but no explicit expression is available.

The principle of offset antennae is described below, the parameters and conditions required to establish the focal point are shown below. A listing of a small BASIC program to do the calculations is shown in Table 1.

Fig 3 shows two views an offset antenna with a symmetrical face. It is a section through a cylindrical paraboloid.

With a little effort, we can prove that:

- Cutting through the surface of the circular paraboloid gives an ellipse
- Projecting the cut onto a plane running at right angles to the axis




Fig 1: Definition of an offset antenna.




Fig 2: Parameters defining an offset antenna.




Fig 3: Two view of an offset antenna with a symmetrical face.

generates a circle.

These important findings make it possible to establish the correction angle for the edge of the offset antenna:

$$\omega = \arctan u \quad (1)$$

where:

$$u = \sqrt{\left(\frac{a}{b}\right)^2} + 1 \quad (2)$$

An equation can also be derived for the coordinates:

$$\frac{u}{b+x_1} - \frac{x_1 u + \frac{a}{b} t \pm \sqrt{\left(x_1 u + \frac{a}{b} t\right)^2 - (x_1 u)^2}}{x_1^2} = \phi \quad (3)$$

The position of the focal point is given by:

$$f = \frac{\frac{b}{2} + x_1}{\sqrt{\left(\frac{a}{b}\right)^2 - 1}}$$

and:

$$l_1 = \sqrt{x_1^2 + (f - z_1)^2} \quad \text{where} \quad z_1 = \frac{x_1^2}{4f}$$

$$l_2 = \sqrt{x_2^2 + (f - z_2)^2} \quad \text{where} \quad z_2 = \frac{x_2^2}{4f}$$

The correction angle can be derived from equations (1) and (2) if it has not already been given by the manufacturer. It is a simple method to calculate it from the reflector width.

It is more difficult to solve equation (3). It can be solved using the small BASIC program, or with a programmable calculator.

First we must delineate the areas on the x axis that gives an unambiguous solutions to equation (3) for $f > \phi$ together with $z_2 > z_1$. Fig 4 shows the interval selection for the solution of the equation by means of interval halving.

We begin in the range $x < \phi$, which is the scope of the two intervals above. If no solution is found, the search is continued in the range $x > \phi$.

It is advantageous to select the interval $\delta x - b$. It can be shown that in (3) a (+) in




Fig 4: Aid to solution of equation (3).




Fig 5: Optimised offset antenna without symmetrical faces.

front of the root for $x > \phi$ never leads to a solution.

If no solution has been found, the areas outside the brackets, which have so far been established as the standard accuracy, $G = 0.001$ (in the program),

$\delta x = \text{accuracy}$, $b = DX$ (in the program) or $N = 1(\delta x - Nb)$, are expanded.

We now select a $G < 0.001$ or $N > 1$. The coordinate x_1 is established with an accuracy of $\pm \delta x$.

In order to obtain a narrower major lobe in the polar diagram of an offset antenna, antennae are manufactured which do not have a symmetrical face (Fig 5). These antennae also have a plane of symmetry $y = \phi$, $2a$ and t can be measured in this plane.

The interval $2b$ can be measured in a plane which runs vertically to the plane $y = \phi$ that contains the points (x_1, z_1) and (x_2, z_2) . To put it another way, for any even section selected through the surface of a rotation paraboloid, the points (x_1, z_1) and (x_2, z_2) can be obtained, which make it possible to measure the lengths $2a$, $2b$ and t .

All the parameters can be measured on a




Fig 6: Examples of offset antennae with the same focal length.



Table 1: BASIC program listing.

```
10 REM OFFSETANTENNE FOCAL POINT
20 REM J.OTYPKA, 1987
30 PI=3.141593
40 INPUT "2A (MM) = ";A
50 INPUT"2B(MM) = ";B
60 INPUT " T (MM) = " ;T
70 REM SYMMETRICAL PARABOLIC
  ANTENNA ?
80 IF A<>B THEN GOTO 130
90 X1=-A/2: X2=A/2
100 F=A^2/16/T
110 Z1=T: Z2=T
120 OM=0: GOTO 710
130 REM ACCURACY G: DX=G*B/2
140 REM STANDARD G=.001
150 INPUT"STANDARD ACCURACY (Y/N)?" ;AO
160 REM X1MAX=N*B/2
170 REM STANDARD N=1
180 INPUT "N STANDARD (Y/N)?" ;BO
190 DX=.001*B/2
200 X1MAX=B/2
210 IF AO="J" THEN GOTO 240
220 INPUT"ACCURACY(1)=" ;G
230 DX=G*B/2
240 IF BO="J" THEN GOTO 270
250 INPUT " N (1) = " ; N
260 X1MAX=N*B/2
270 REM OUTPUT PARAMETERS
280 AB=A/B
290 U=SQR(AB^2-1)
300 REM CORRECTION ANGLE
310 OM=ATN(U)*18000/PI
320 OM=INT(OM)/100
330 POM=AB*T
340 X1MIN=-.5*POM/U+DX
350 X0MIN=-B/2+DX
360 IF X1MIN > X0MIN THEN GOTO 380
370 X1MIN=X0MIN
380 DEF FN F(X,K)=U/(B/2+X)-(X*U+POM+
  K*SQR((X*U+POM)^2-X*X*U^2))/(X*X)
390 REM SOLUTION EXISTS FOR X1<0 ?
400 K=3
410 K=K-2
420 DMIN=FN F(X1MIN,K)
430 D0=FN F(-DX,K)
440 IF SGN(DMIN)=SGN(D0) AND K=1
  THEN GOTO 410
450 IF SGN(DMIN)=SGN(D0) THEN GOTO 530
460 X1=X1MIN
470 F1=DMIN
480 X2=-DX
490 F2=D0
500 GOSUB 990
510 X1=X3
520 GOTO 660
530 X1=DX
540 FL=FN F(X1,-1)
550 X2=X1MAX
560 F2=FN F(X2,-1)
570 IF SGN(F1)<>SGN(F2) THEN GOTO 620
580 AO="N"
590 BO="N"
600 PRINT "EXPAND LIMITS!"
610 GOTO 210
620 REM SOLUTION FOR X1>0
630 K=-1
640 GOSUB 990
650 X1=X3
660 REM PARAMETERS SOUGHT
670 F=(B/2+X1)/2/U
680 Z1=X1*X1/4/F
690 X2=X1+B
700 Z2=X2*X2/4/F
710 L1=SQR(X1*X1+(F-Z1)*(F-Z1))
720 L2=SQR(X2*X2+(F-Z2)*(F-Z2))
730 DZ=(L1^2+L2^2-A^2)/2/L1/L2
740 REM ACS:750-800
750 IF DZ<> THEN GOTO 770
760 DZ=90:GOTO 810
770 ACS=ATN(SQR(1/DZ/DZ-1))
780 IF DZ>0 THEN GOTO 800
790 ACS=3.141593-ACS
800 DZ=INT(ACS*18000/PI)/100
810 PRINT
820 PRINT "OFFSET ANTENNA
  PARAMETERS:"
830 PRINT
840 PRINT "2A (MM) ;A
850 PRINT"2B (MM) = ;B
860 PRINT"DEPTH (MM) = ";T:PRINT
870 PRINT "CORRECTION ANGLE
  (DEGREE) = " ;OM
880 PRINT"FOCAL POINT LENGTH F (MM) = F
890 PRINT"X1 (MM) = " ;X1
900 PRINT"Z1 (MM) = " ;Z1
910 PRINT"X2 (MM) = " ;X 2
920 PRINT"Z2 (MM) = " ;Z2
930 PRINT"L1 (MM) = ";L1
940 PRINT"L2 (MM) = ";L2
950 PRINT"ANGLE 2DZ (DEGREE) = "DZ
960 PRINT
970 STOP
980 REM INTERVAL HALVING
990 X3=(X1+X2)/2
1000 DF=ABS(X1-X2)
1010 IF DF<DX THEN GOTO 1100
1020 F3=FN F(X3,K)
1030 IF SGN(F3)<>SGN(F1) THEN GOTO 1070
1040 X1=X3
1050 F1=F3
1060 GOTO 990
1070 X2=X3
1080 F2=F3
1090 GOTO 990
1100 RETURN
```



real offset antenna. There remains only the problem of the correct classification of points (x_1, z_1) and (x_2, z_2) , which are in the plane of symmetry.

In theory the classification is clear, but in practice there are always problems, so manufacturers data would be a help.

The point (x_1, z_1) lies at a position on the edge where the curvature of the section in the plane $y = \phi$ is greater than the curvature at point (x_2, z_2) .

Fig.6 shows 5 examples offset antennae. The focal length of 35mm. is selected, and then some calculations are carried out.

Ex 1 Using the BASIC program, we must initially increase the intervals after entering the standard parameters. The standard accuracy has been retained, N has been defined as $N = 2$, and finally $N = 3$ has been selected, which led to a solution.

Ex 2 It was not necessary to alter the standard parameters.

Ex 3 The section runs through the impermissible area $x_1 = \phi$, the intervals had to be increased again. So $G = 0.00001$ was selected and x_1 calculated. The area around $x_1 = \phi$ can be avoided if the input parameter is altered "infinitesimally". In practice, this case presents no difficulties, and so no special researches or calculations were needed.

Ex 4 A higher accuracy was selected for x_1 , from $x = 0.1\text{mm}$ to $x = 0.01\text{mm}$. If we lay down that $2a = 2b$, we are presetting the values of a symmetrical parabolic antenna, which is shown in Fig.6.

Here are some applications for the program:

- Establishing focal point:
 - For correct placing of radiator
 - Determining phase centre of radiator used
- Modification or copying a parabolic antennae

Literature

Urceni ohniska ofsetore anteny Sdelovaci tehnika, Otypka, J, c. 8, 1992